

Traitement de texte

Une fois la fastidieuse opération d'entrée du programme effectuée, sauvez votre programme soit sur cassette, soit sur disquette. La compatibilité est alors automatique.

Faites "RUN" et vous verrez s'afficher à l'écran la question : "Désirez-vous reconfigurer l'imprimante ?" En effet il peut être nécessaire d'affecter les caractères accentués à d'autres codes ASCII que ceux prédéfinis dans le programme.

Si vous répondez oui, vous verrez alors s'afficher chaque caractère à l'écran et il vous faudra affecter à chacun de ces caractères la valeur adéquate. Pour cela reportez-vous au manuel de votre imprimante).

Vous accédez maintenant au menu. Les options suivantes vous sont alors proposées :

1) Chargement

Si vous sélectionnez cette option, le programme chargera d'une cassette, ou d'une disquette le fichier de texte préalablement sauvé. Le fichier est nommé dans tous les cas "texte".

2) Sauvegarde

Cette fonction est identique à celle de chargement en mode sauvegarde.

3) Effacement mémoire

Efface tout le fichier de texte contenu en mémoire. (Attention le format du texte sera également effacé).

4) Azerty

Passage en mode Azerty

5) Qwerty

Retour en mode Qwerty

6) Format des pages

Définit le nombre de pages et le nombre de colonnes du texte. (Cette option doit absolument être sélectionnée avant de taper du texte).

7) Copie sur imprimante

Copie le texte sur imprimante dans la largeur définie par l'option de format.

8) Accès au texte

Permet de passer au mode d'édition du texte.

Les caractères de contrôle :

DEL : efface tout caractère présent devant le curseur.


Les curseurs :

Ils permettent de se déplacer dans le texte.

ENTER : permet de passer à la ligne suivante directement, la ligne où est situé le curseur.

CTRL+c : retour au menu

Eric Charton



```
20 '
30 '      traitement de texte
40 '
45 '
50 OPENOUT "ID":CLOSEOUT
80 a=64:a1=92:a2=93:a3=94:a4=123:a5=124:a6=125
100 WIDTH 80
110 GOSUB 60000
115 MODE 2
120 INPUT "desirez vous reconfigurer l'imprimante ",a$
125 IF a$="o" THEN GOSUB 62000
200 GOTO 30000
505 '
507      'INITIALISATION
510 '
520 X=1
```




```
530 Y=1
535 ec=1
540 CLS
550 MODE 2
1000 '
1010 ' ROUTINE GENERALE DE TEST CLAVIER
1020 '
1050 GOSUB 5000
1060 GOSUB 7000
1100 A$=INKEY$
1105 IF A$<>" THEN IF ASC (A$)>239 THEN GOSUB 2000:GOTO 1100
1110 IF A$="" THEN 1100
1116 IF ASC (a$)=127 THEN GOSUB 2140:GOTO 1100
1117 IF (a$)=CHR$ (13) THEN GOSUB 3000:GOTO 1100
1118 IF a$="" THEN 30000
1120 LOCATE Y,X:PRINT A$:tr$ (x+(24*ec),y)=a$: Y=Y+1:IF Y>LONG THEN Y=1:X=X+1
1130 GOSUB 7000
1150 IF x>24 THEN ec=ec+1:GOSUB 5000:x=1:y=1
1200 GOTO 1100
2000 ' saisie des caracteres de controle
2100 IF INKEY$=CHR$ (243) AND Y<long THEN Y=Y+1
2110 IF INKEY$=CHR$ (242) AND Y>1 THEN Y=Y-1
2120 IF INKEY$=CHR$ (240) THEN X=X-1:IF x=0 THEN IF ec>1 THEN ec=ec-1:GOSUB 500
0:x=24:y=1 ELSE x=1
2130 IF INKEY$=CHR$ (241) THEN X=X+1 :IF x=25 THEN CLS :ec=ec+1:GOSUB 5000:x=1:
y=1
2140 IF INKEY (79)=0 THEN tr$(x+(24*ec),y)="" :IF Y>1 THEN y=y-1 ELSE IF X>1 THEN
x=x-1:y=long:LOCATE 1,x+1:PRINT " "
2150 GOSUB 7000
2200 RETURN
3000 IF x<24 THEN LOCATE y,x:PRINT tr$(x+(24*ec),y):IF tr$(x+(24*ec),y)="" THEN
LOCATE y,x:PRINT " "
3005 IF x<24 THEN x=x+1:y=1
3010 GOSUB 7000
3100 RETURN
5000 '
5003 ' increment d'une page
5004 '
5010 IF ec=page+1 THEN y=long:x=24:ec=ec-1: RETURN
```


LISTING

```
5015 CLS
5020 GOSUB 6000
5030 FOR t=1 TO 24
5040 FOR u=1 TO long
5050 LOCATE u,t
5060 PRINT tr$(t+(24*ec),u)
5070 NEXT u
5080 NEXT t
5090 RETURN
6000 '
6010 ' indications
6020 '
6030 LOCATE 1,25:PRINT "page no :";ec;"    largeur :";long;"    nombre de pages:"
;page
6100 RETURN
7000 '
7010 ' affichage curseur
7020 '
7040 IF y>1 THEN LOCATE y-1,x:IF tr$(x+(24*ec),y-1)=" " THEN PRINT " " ELSE PRINT
    tr$(x+(24*ec),y-1)
7050 IF y<long THEN LOCATE y+1,x:IF tr$(x+(24*ec),y+1)=" " THEN PRINT " " ELSE PR
    INT tr$(x+(24*ec),y+1)
7055 IF x=0 OR y=0 THEN GOTO 55000
7060 IF x>1 THEN LOCATE y,x-1:IF tr$(x-1+(24*ec),y)=" " THEN PRINT " " ELSE PRINT
    tr$(x-1+(24*ec),y)
7070 IF x<24 THEN LOCATE y,x+1:IF tr$(x+1+(24*ec),y)=" " THEN PRINT " " ELSE PRIN
    T tr$(x+1+(24*ec),y)
7100 PRINT CHR$(22)+CHR$(1)
7200 LOCATE y,x
7320 PRINT CHR$(210)
7330 PRINT CHR$(22)+CHR$(0)
7400 RETURN
9075 IF TR$(I,J)=" " THEN GOTO 59085
20000 '
30000 '
30010 '
30015 '    options
30030 '
30040 '
```




```
30045 ON ERROR GOTO 55000
30050 MODE 1
30060 LOCATE 10,2:PRINT "options
30070 LOCATE 3,5:PRINT "1 chargement":LOCATE 3,6:PRINT "2 sauvegarde":LOCATE 3,7

30080 PRINT "3 effacement memoire":LOCATE 3,8:PRINT"4 azerty":LOCATE 3,9:PRINT"5
  qwerty": LOCATE 3,10:PRINT "6 format des pages":LOCATE 3,11:PRINT "7 copie sur
  imprimante":LOCATE 3,12:PRINT "8 acces au texte"
30199 a$=INKEY$
30200 IF a$="" THEN 30199
30210 IF a$="8"THEN 505
30215 IF a$="7"THEN 59000
30220 IF a$="6"THEN GOTO 32000
30225 IF a$="3"THEN CLEAR:GOTO 30000
30230 IF a$="4"THEN GOSUB 50000 :PRINT "azerty OK":FOR i=1 TO 1000:NEXT:GOTO 300
  00
30240 IF a$="5"THEN GOSUB 52000 :PRINT "qwerty OK":FOR i=1 TO 1000:NEXT:GOTO 300
  00
30250 IF a$="1" THEN GOTO 56000
30260 IF a$="2" THEN GOTO 57000
31999 GOTO 30199
32000 LOCATE 2,23
32005 CLEAR
32010 INPUT "nombre de pages (1 page=24 lignes)?",page
32100 INPUT "largeur d'une page (10/80)",long
32105 DIM tr$((page+1)*25,long+1)
32110 GOTO 30000

50000 '
50010 '
50020 '      azerty
50030 '
50040 '

51000 KEY DEF 67,0,97,65:REM transforme q en a
51100 KEY DEF 59,0,122,90:REM transforme w en z
51200 KEY DEF 69,0,113,81:REM transforme a en q
51300 KEY DEF 71,0,119,87:REM transforme z en w
51400 KEY DEF 29,0,109,77:REM transforme : en m
51500 KEY DEF 38,0,58,42:REM transforme m en :
51600 RETURN
```




```

52000 '
52100 '
52200 '      qwerty
52210 '
52220 '
52300 KEY DEF 67,0,113,81:REM transforme a en q
52310 KEY DEF 59,0,119,87:REM transforme z en w
52320 KEY DEF 69,0,97,65:REM transforme q en a
52330 KEY DEF 71,0,122,90:REM transforme w en z
52340 KEY DEF 29,0,58,42:REM transforme m en :
52350 KEY DEF 38,0,109,77:REM transforme : en m
53000 RETURN
55000 '
55010 '
55020 '  tableau non defini
55030 '
55040 '
55050 MODE 1
55060 LOCATE 24,5
55070 PRINT "format des feuillets  non defini erreur"
55075 IF INKEY#="" THEN 55075
55080 GOTO 30000
56000 '
56010 '  CHARGEMENT
56020 '
56025 CLEAR
56030 OPENIN "texte"
56040 INPUT #9,page:INPUT #9,long
56045 DIM tr$((page+1)*24,long )
56050 FOR I =0 TO ((PAGE+1)*24):FOR J=0 TO LONG
56060 LINE INPUT #9,tr$ (i,j)
56070 NEXT J:NEXT i
56080 CLOSEIN
56090 GOTO 30000
57000 '
57010 '  SAUVEGARDE
57020 '
57030 OPENOUT "TEXTE"
57035 PRINT #9,page:PRINT #9,long
57040 FOR I =0 TO (( PAGE+1)*24):FOR J=0 TO LONG
57050 PRINT #9,tr$ (i,j)
57060 NEXT J:NEXT i
57065 CLOSEOUT
57070 GOTO 30000
59000 '
59010 '
59020 '
59030 '  sortie sur listing
59040 '
59050 '
59060 '
59070 FOR I =25 TO(24*(PAGE+1)):FOR J=1 TO LONG
59075 IF TR$(I,J)="" THEN PRINT #8," " :GOTO 59085
59080 PRINT #8,tr$(I,J);
59085 NEXT J:PRINT #8,CHR$(13):NEXT i
59090 GOTO 30000
60000 '
60010 '
60020 '      caracteres accentues
60030 '
60040 '
60100 SYMBOL AFTER 64
60110 SYMBOL a,&60,&10,&78,&C,&7C,&CC,&76
60120 SYMBOL a1,&0,&0,&3C,&66,&60,&3E,&8,&18
60130 SYMBOL a2,&1C,&32,&38,&6C,&38,&10,&9D,&6D
60140 SYMBOL a3,&18,&24
60150 SYMBOL a4,&6,&8,&3C,&66,&7E,&6D,&3C
60160 SYMBOL a5,&3D,&8,&66,&66,&66,&66,&3E
60170 SYMBOL a6,&6D,&1D,&3C,&66,&7E,&6D,&3C
60180 KEY 128,CHR$(a6)
60190 KEY 129,CHR$(a)
60200 KEY 131,CHR$(a2)
60210 KEY 132,CHR$(a1)
60220 KEY 134,CHR$(a3)
60230 KEY 135,CHR$(a4)
60240 KEY 137,CHR$(a5)
60250 KEY 136,CHR$(32)
60255 KEY 133,CHR$(32)
60260 KEY 130,CHR$(32)
60300 RETURN
62000 CLS
62005 PRINT CHR$(a);:INPUT ": correspondance ";a
62010 PRINT CHR$(a1);:INPUT ": correspondance ";a1
62020 PRINT CHR$(a2);:INPUT ": correspondance ";a2
62030 PRINT CHR$(a3);:INPUT ": correspondance ";a3
62040 PRINT CHR$(a4);:INPUT ": correspondance ";a4
62050 PRINT CHR$(a5);:INPUT ": correspondance ";a5
62060 PRINT CHR$(a6);:INPUT ": correspondance ";a6
62100 GOSUB 60000
63000 GOTO 30000

```